
1. BEVEZETÉS 

A GAMS az 1980-as évek vége óta a nyugat-európai és észak-amerikai gazdasági 
jellegő felsıoktatási intézményekben döntési problémák modellezésére széleskörően 
használt, magasszintő, matematikai modellezı nyelv. 
 
A nemzetközi szakirodalomban általános nézet szerint a gazdasági felsıoktatásban is 
elterjedt, hagyományosan strukturált (Fortran, Pascal, C, Basic stb.) és 
objektumorientált (Visual Basic, C++, Java, Delphi stb.) programnyelvek mellett 
érdemes a magasszintő, matematikai modellezı nyelvek oktatására nagyobb 
hangsúlyt helyezni, mivel ez utóbbiak az önálló, közgazdasági modellezést már 
viszonylag rövid tanulás után jelentısen megkönnyítik. A GAMS nyelv alapjainak 
megismertetése tehát nem tőnik haszontalannak, különösen igaz ez az informatikai 
tárgyakat magasabb óraszámban hallgató diákokra, hiszen a döntéstámogatás 
minden gazdasági informatikai tantárgyprogramban kiemelt szerepet játszik. 
 
Mielıtt rátérnénk a GAMS nyelv sajátosságaira, érdemes röviden tisztázni azt, hogy 
a mit értünk szakértıi rendszer és döntéstámogató rendszer alatt. 
 
A szakértıi rendszerek olyan intelligens rendszerek, amelyek egy adott területen 
képesek emberi szakértıket kiváltani vagy részben helyettesíteni, s szakértıi 
döntéseket támogatni illetve automatizálni. Felhasználásuk igen sokrétő, többek 
között sikeresen alkalmazhatók az alábbi területeken: 

• diagnózisok készítése – betegségek felismerése 
• hibakeresés – mechanikai rendszerek hibás alkatrészeinek megtalálása 
• adatok interpretálása – kémiai molekulák szerkezetének meghatározása 
• elırejelzések készítése – projekttervezés 
• oktatás és tréning - adótanácsadás 

 
A szakértıi rendszerek általában az alábbi komponenseket tartalmazzák: 

• tudásbázis, azaz HA … AKKOR típusú szabályok összessége, 
• adatbázis, amely egy adott terület ismereteit, tényeit tartalmazza, 
• következtetı modul (inference engine), amely tények és szabályok 

felhasználásával következtetéseket képes levonni, 
• magyarázó modul, amely megmagyarázza, hogy a szakértıi rendszer hogyan 

jutott egy bizonyos következtetésre, azaz milyen következtetéssorozat vezetett 
az eredményhez, 

• tudásszervezı modul, amely új információkat képes győjteni az emberi 
szakértıktıl, illetve külsı adatbázisokból, 

• felhasználói illesztés, amely az ember – gép kommunikációt végzi. 
 
A döntéstámogató rendszerek (DTR) olyan interaktív rendszerek, amelyek mind az 
egyéni, mind a csoportos döntéshozatalt támogatják, amikor a probléma csak részben 
vagy egyáltalán nem strukturált. 

A döntéstámogató rendszerek legfontosabb feladatai az alábbiak: 
• modellalkotás, 
• érzékenységvizsgálat, 
• célérték keresése, 
• kockázatelemzés, 
• grafikus elemzés. 

 
A döntéstámogató rendszerek legfontosabb jellemzıi: 

• strukturálatlan vagy kevéssé strukturált problémák megoldására alkalmasak, 
• kifinomult elemzı és modellezı eszközöket használnak, 
• könnyen adaptálhatók, rugalmasak. 

 
A GAMS (General Algebraic Modeling System) programnyelvet a Világbank 
megbízásából Alexander Meeraus és munkatársai fejlesztették ki az 1980-as évek 
végén. A döntéstámogatáson belül elsısorban klasszikus operációkutatási 
feladatokra használják szerte a világon, de a különbözı, ma már a Világhálón is 
hozzáférhetı modellkönyvtárakban (pl. http://www.gams.com/modlib) számos, a 
magyar viszonyok között is adaptálható, jórészt a Világbank által készített vagy 
készíttetett modell egyszerősített változatát találjuk elsısorban az alábbi 
témakörökben: 
 

• makrogazdaságtan: optimális növekedés, beruházási modellek, ágazati 
modellek 

• mikrogazdaságtan: kínálati modellek, árdiszkrimináció, piaci helyzetek 
• menedzsment: just-in-time szállítás, munkaerı-gazdálkodás, szervezetek 

kialakítása, szervezeti munka hatékonysága 
• mezıgazdaság és agrárközgazdaságtan: a fejlıdı világ mezıgazdasági és 

élelmezési problémái, általános mezıgazdasági modellek, öntözési és 
talajfelhasználási problémák 

• általános egyensúly: a fejlıdı (pl. India, Kamerun), az utóbbi évtizedekben 
gyors gazdasági fejlıdési pályára állt (pl. a Koreai Köztársaság, Kína) és a 
különbözı okokból az elmúlt idıszakban gazdasági szerkezetváltásra 
kényszerült fejlett országok (pl. Finnország) egyensúlyi problémái, adózási 
kérdések, a versenyt korlátozó piaci helyzetek, globális egyensúlytalanság, a 
gazdasági integrációk által generált egyensúlyi problémák (pl. Nagy 
Britannia) 

• ökonometria: regresszió, maximum likelihood becslés 
• gazdasági fejlıdés: strukturális problémák, szők keresztmetszetek, új 

erıforrások, termelési tényezık helyettesítése 
• energiagazdaságtan: nemzetközi földgáz- és kıolaj-kereskedelem, megújuló 

energiaforrások, a fejlett, de sok energiát fogyasztó ipari országok problémai 
(pl. Koreai Köztársaság, Egyesült Államok), az OPEC optimális árképzése, az 
energiatermelés és felhasználás környezetszennyezı hatásai 

• pénzügy: portfolió-elemzés, pénzügyi tervezés, kockázatkezelés 
• nemzetközi kereskedelem: rézkereskedelem, alumíniumkereskedelem, 

szénhidrogének kereskedelme, észak-déli kereskedelem és tıkeáramlás 

• mőszaki tudományok: vegyészet, csıvezetékek és szállítási útvonalak 
tervezése, alkatrészszállítás, vízi utak biztonságos hasznosítása, jármővek 
szárnytervezése, optimális felületek kialakítása, árvízvédelem, szivattyú 
hálózatok tervezése, hulladékanyag-minimalizálás 

• erdıgazdálkodás: fakitermelés és erdıtelepítés, az erdıgazdálkodás 
környezeti terhei, a trópusi és az északi erdık megóvása 

• sztochasztikus programozás: optimális beruházás, repülıjáratok tervezése 
• statisztikai, matematikai alkalmazások 

 
A fent említett példák is mutatják, hogy a GAMS meglehetısen széles körben 
használható programnyelv, azonban az összetettebb modellek kialakítása elég 
bonyolult, a legtöbb modell leegyszerősített változatának bemutatása is meghaladná 
az egyetemi oktatás kereteit, kezdetnek azonban néhány egyszerőbb modell 
áttekintése is elegendı. Természetesen a GAMS programnyelv - a világbanki 
preferenciák ellenére - teljesen semleges fejlesztıeszköz, mindegy, hogy - a 
közgazdasági modellezésnél maradva - milyen közgazdasági iskola gondolatain 
nyugvó modellt kívánunk a segítségével megfogalmazni.  
 
A jegyzetben használt GAMS 20.7-es Student (leegyszerősített, kevesebb modellt és 
változót használó) változata a washingtoni GAMS Development Corporation 
hivatalos oldaláról regisztráció után ingyenesen letölthetı (http://www.gams.com), 
de a cég giesseni európai központjának a honlapján is elérhetı. 
(http://www.gams.de) Az ingyenes változat is rendkívül hasznos fejlesztıeszköz, 
jómagam a Világhálón nagyon sok, egymástól rendkívül eltérı fejlettségő és 
kultúrájú ország egyetemein és kutatóintézeteiben megírt GAMS kódra találtam. (Pl. 
Svédország, Irán, Németország, Kazahsztán, Spanyolország, Törökország - 
meglehetısen sokszínő országcsoport.) A GAMS „ingyenessége” ellenére a magyar 
honlapokon 2003 elején a legismertebb keresıprogramok segítségével sem sikerült 
GAMS modellekre találnom, bár történtek arra utalások, hogy a GAMS nyelvet itt-ott 
egy-két órában oktatják. 
 
A jegyzet elsı felében egy példaalkalmazáson keresztül rövid, de a teljességre 
törekvı áttekintést nyújtok a GAMS programnyelv legfontosabb jellegzetességeirıl. 
A GAMS iránt mélyebben érdeklıdı olvasók Richard E. Rosenthal GAMS User's 
Guide c. mővében teljes leírást találnak, azonban a példaalkalmazás megértése után 
bárki képessé válhat a különbözı, gyakrabban elıforduló lineáris programozási 
feladatok gyors megoldására anélkül, hogy a meglehetısen sokrétő GAMS 
programnyelv használatát teljes egészében elsajátítaná. 
 
A jegyzet második felében néhány jól ismert operációkutatási alkalmazáson kívül 
bonyolultabb közgazdasági modellek leegyszerősített változatai közül mutatok be 
néhányat. 

2. SZÁLLÍTÁSI PROBLÉMA 

Példaalkalmazásunk a lineáris programozás Dantzig munkássága nyomán 
klasszikussá vált szállítási problémája, amely az optimalizációs technológia 
fejlıdésében betöltött szerepe és a gyakori hivatkozások révén az operációkutatási 
feladatok "állatorvosi lovának" is tekinthetı. A GAMS és a hozzá hasonló 
matematikai modellezı nyelvek erısségeinek illusztrálására is kiválóan alkalmas, 
mivel a szállítási probléma, függetlenül az adott modell nagyságától, egyszerő, 
átlátható algebrai szerkezettel rendelkezik. A késıbbiekben majd nyilvánvalóvá 
válik, hogy nagyobb és bonyolultabb szállítási probléma esetén, az egyszerőbb 
szállítási feladatra vonatkozó, a GAMS bemeneti fájljában megfogalmazott állításaink 
alig változnak. 
 
A klasszikus szállítási problémánál megadjuk egy árucikk különbözı gyárakban 
termelt és a helyi piacokon keresett mennyiségét, ezenkívül az egyes gyárak és 
piacok közötti szállítási egységköltségek is adottak. Hatékonysági kérdésünk a 
következı: Mennyit szállítsunk az egyes gyárak és piacok között úgy, hogy a 
szállítási összköltség minimális legyen? 
 
A probléma matematikai modelljét az alábbi formában írhatjuk le: 
 
Halmazok: 
  i = gyárak 
  j = piacok 
 
Exogén változók: 
  ai = i gyár árukínálata (szállítóládában megadva) 
  bi = j piac árukereslete (szintén szállítóládában megadva) 

cij = egységnyi szállításra esı költség i gyár és j piac között (pénzegység/láda) 
 
Döntési (endogén) változók: 

xij = szállításra kerülı árumennyiség i gyár és j piac között (szállítóládában) 
  ahol xij ≥ 0, minden i,j-re 
 
Korlátok: 
  Kínálati korlát i gyárban:  ∑j  xij  ≤ ai, minden i-re  
  Kielégítendı kereslet j piacon:  ∑i xij  ≤ bj, minden j-re  
 
Célfüggvény: 
  ∑i∑j cijxij → minimum 
 
A fenti matematikai modellen alapuló GAMS példánk megmutatja a GAMS 
modellek néhány olyan alapvetı követelményét, amelyet a gyakorlat során érdemes 
mindig betartani: 
 
1. A modell elemeinek meghatározása és csoportosítása típusok szerint történik. 
2. Az elemeknél be kell tartanunk azt a szabályt, hogy semmire sem hivatkozhatunk 

anélkül, hogy azt korábban ne definiáltuk volna. 
3. Az elemek összetevıi kötöttek, tehát meghatározott rend szerint kell ıket 

megadnunk. 

4. A számadatokat, amelyeket a GAMS megoldó algoritmusai értelmeznek, lehetıleg 
kis abszolút nagyságrendben kell megadnunk. (Pl. ezer forintban.) 

 
A modellezéssel foglalkozó szakemberek az egyes típuselemekre gyakran különbözı 
elnevezéseket használnak. Például a modell ismert adatait, együtthatóit, paramétereit 
exogén változóknak, a döntési változókat endogén változóknak is nevezhetjük. A 
GAMS által alkalmazott terminológia a következı: A halmazokat sets, a 
paramétereket parameters, a döntési változókat variables, a korlátokat és a 
célfüggvényeket pedig equations néven azonosítjuk.  
 
A szállítási példa GAMS modellje külsıleg nagyon hasonlít a fenti matematikai 
modellre. Lényegében a matematikai modell kis átalakításával a fordító által 
értelmezhetı kódhoz jutunk. 
 
A szállítási problémára vonatkozó példánkban két konzervgyárból három helyi 
piacra szállítunk. A példa azonos Dantzig 1963-ban publikált, mára már klasszikussá 
vált szállítási feladatával, azonban a modellt kissé „magyarítottuk”. A modell exogén 
változóit a következı táblázat mutatja:  
 

 Szállítási távolság  
 Piacok  
Gyárak Sopron Szombathely Zalaegerszeg Kínálat 
 
Kecskemét 

 
302 

 
279 

 
266 

 
350 

Szeged 
 

308 354 342 600 

Kereslet 325 300 275  
 
A szállítási távolságot kilométerben adtuk meg, a szállítási költségrıl pedig azt 
feltételezzük, hogy egységesen 90 forint /szállítóláda /kilométer. A probléma GAMS 
kódja a következı: 
 
Sets 

i kgyarak / Kmet, Szeged / 
j piacok / Sopron, Szhely, Zeg /; 

Parameters 
a(i) i konzervgyar kapacitasa ladaban 

/  Kmet 350 
Szeged 600 / 

b(j) kereslet j piacon ladaban 
/  Sopron 325 

Szhely 300 
Zeg 275 /; 

Table d(i,j) tavolsag kilometerben 
Sopron   Szhely  Zeg 

Kmet   302   279   266 
Szeged   388   354   342 ; 
 
Scalar f kilometerre eso fuvardij ladankent forintban /90/; 

Parameter c(i,j) ladankenti szallitasi koltseg forintban; 
 
c(i,j) = f*d(i,j); 

 
Variables 

x(i,j) szallitott mennyiseg ladaban 
z forintban mert osszes szallitasi koltseg ; 

 
Positive variable x ; 
 
Equations 
 
cost celfuggveny 
supply(i) i gyar kinalati korlatja 
demand(j) j piac kielegitendo kereslete; 
 
cost.. z =e= sum((i,j), c(i,j)*x(i,j)) ; 
supply(i) .. sum(j, x(i,j)) =l= a(i) ; 
demand(j) .. sum(i, x(i,j)) =g= b(j) ; 
 
Model szallfel /all/ ; 
 
solve szallfel using lp minimizing z ; 
 
display x.l, x.m ; 
 

 
   
  2.1 ábra A szállítási modell bemeneti fájlja GAMSIDE 20.7 Student felületen 

3. A GAMS MODELLEK ÁLTALÁNOS SZERKEZETE 

A jegyzet hátralévı részében, a GAMS alapvetı komponenseit elemezzük, s minden 
elemnél visszautalunk a már megoldott szállítási példánkra. 
 
Az input fájl részei Az output fájl részei 

• Sets 
             Deklaráció 
             A tagok felsorolása, meghatározása 
• Adatok 

(Parameters, Tables, Scalar) 
      Deklaráció 
      Értékadás 

• Variables 
                     Deklaráció 
                     Típus megadása 

• Korlátok és/vagy kezdıértékek 
megadása (nem kötelezı) 

• Equations 
              Deklaráció 
              Definíció 
• Model és Solve utasítás 
• Display utasítás (nem kötelezı)  
 

• Echo Print (visszajelzés) 
• Referencia térkép 
• Egyenletlisták 
• Státusz jelentések 
• Eredmények 

 
 

 
A fentieken kívül az input fájl több, a professzionális alkalmazásoknál fontos 
szerepet játszó, választható összetevıt is tartalmaz (hamis adatok ellenırzése, az 
eredmények testreszabott megjelenítése, korábbi modellek mentése és helyreállítása, 
modelltöbbszörözés stb.), azonban a jegyzet terjedelmi korlátai csak az alapvetı 
összetevık bemutatását teszik lehetıvé.   
 
Mielıtt az egyes alkotóelemeket részletesen bemutatnánk, néhány általános érvényő 
észrevételt tennénk: 
 
1. A GAMS modell GAMS nyelvben megfogalmazott utasítások győjteménye. Az 

utasítások sorrendjénél egy szabályt mindig be kell tartani: a modell egyetlen 
elemére sem hivatkozhatunk az elem elızetes deklarációja nélkül! 

2. A GAMS utasításokat tipográfiailag szabadon megfogalmazhatjuk. Egy utasítást 
több sorba, több utasítást egy sorba is írhatunk, azonban tapasztalataink szerint a 
standard angol szavak karakterkészletétıl jobb nem eltérni, pl. az ékezetes betők 
hibához vezethetnek. 

3. A kezdı GAMS felhasználók jobban teszik, ha minden egyes utasítást pontos 
vesszıvel zárnak le, mint ahogy az a példaalkalmazásban is látható. 

4. A GAMS fordító nem tesz különbséget a kis-, és nagybetős szavak között, 
bármelyik forma szabadon használható. 

5. A megfelelı dokumentáció a matematikai modellek lényeges és hasznos eleme. A 
magyarázatokat, megjegyzéseket szerencsésebb a modellbe ágyazni, mintsem 
külön lejegyezni. A GAMS modellben alapvetıen két lehetıség is van a 
magyarázatok beágyazására. A GAMS fordítója figyelmen kívül hagy minden "*" 
jellel kezdıdı sort, pontosabban megjegyzésként kezeli azokat. A másik, talán 

hasznosabbnak tőnı módszer a magyarázatok GAMS utasításokba történı 
beszúrása. A szállítási modell kisbetős szavai a második módszert szemléltetik. 

6. Ahogy az input fájl elemeinek felsorolásánál is látható, a GAMS elemek 
létrehozása két részbıl áll: deklarációból és definícióból (értékadásból, 
felsorolásból). A deklaráció itt létrehozást, elnevezést-megnevezést jelent, a 
definíció során pedig a létrehozott és névvel ellátott "objektumot" értékkel, 
formával látjuk el. Az equations elemtípusnál a deklarációt és a definíciót külön 
GAMS utasításokkal kell elvégezni. A többi elemtípusnál a deklarációt és a 
definíciót egy utasításban is elvégezhetjük. 

7. Az egyes típuselemek, entitások elnevezéseinek betővel kell kezdıdniük, s 
maximum tíz karakterbıl (betőkbıl és számokból) állhatnak. 

 

4. SETS 

A GAMS modellek alapvetı építıeleme a Set, amely egyértelmően megfeleltethetı az 
algebrai modellek halmazaival. A set-et tömbként is fordíthatnánk, azonban a 
Parameters vagy a Tables utasítással is tömböket hozunk létre, ezért a fogalmi 
zavarok elkerülése érdekében maradjunk a „halmaz” kifejezésnél, ill. az angol 
eredetinél. Szállítási modellünk csak egy set utasítást tartalmaz: 
 

Sets 
i kgyarak / Kmet, Szeged / 
j piacok / Sopron, Szhely, Zeg /; 

 
Az utasítás eredménye nyilvánvaló. Deklaráltunk két halmazt, és i valamint j névvel 
láttuk el ıket. Az i és a j halmaz elemeit is felsoroltuk, amelyek hagyományos 
matematikai jelöléssel a következık: 
 
 i = {Kecskemét, Szeged} 
 j = {Sopron, Szombathely, Zalaegerszeg}. 
 
A halmaz elemeinek felsorolásánál figyeljük meg a GAMS utasítások és az általános 
matematikai formák közötti különbségeket! A GAMS "/" jeleket használ "{}" helyett, 
mivel a billentyőzetek egy része nem tartalmaz kapcsos zárójelet. A tagok nem 
állhatnak több szóból, kötıjelet kell használnunk az összefőzésükre.  
 
A fenti sets utasításban szereplı kisbetős szavakat a GAMS terminológia szerint text-
nek nevezzük. A text megadása nem kötelezı, magyarázatként szolgál, a modell 
kiegészítı része. A GAMS fordító a text részt nem fordítja le, csak elmenti. 
 
Az egyes set elemtípusokat nem kötelezı egyetlen sets-be sőríteni. A 
példaalkalmazás sets utasítását így is megfogalmazhatjuk: 
  

Set i kgyarak   /Kmet, Szeged/; 
Set j piacok     /Sopron, Szhely,  Zeg/; 

 

Az üres helyek nagysága a kis-, és nagybetőkhöz hasonlóan szabadon 
megválasztható. A GAMS felhasználók, mint minden programozó, általában saját 

stilisztikai szabályokat követnek. Pusztán az egyéni ízléstıl függ az is, hogy több set-
et, vagy egyetlen sets-et használunk, a GAMS nem tesz különbséget a két forma 
között. 
 
A halmazelemek legfeljebb 10, a magyarázatul szolgáló részek legfeljebb 80 
karakterbıl állhatnak. (Emiatt kénytelen voltam a magyar településnevek rövidítéseit 
használni.) 
 
Abban az esetben, ha a halmaz elemei szekvenciát alkotnak, a tagok felsorolásánál a 
"*" jel hasznos segítséget jelenthet: 
 

Set t ido periodus  /1991*2000/; 
Set g gepek  /gep1*gep24/; 

 
Algebrai jelöléssel: 
 

t = {1991, 1992, 1993, ..., 2000} 
g = {gep1, gep2, gep3, ..., gep24}. 

  
Meg kell jegyezni, hogy a halmaz elemei karaktersorozatként (sztringként) 
tárolódnak, ezért pl. t elemei sem szám típusúak.  
 
Az alias utasítás segítségével egy korábban már deklarált halmazt  új névvel 
láthatunk el. 
 
A példáink során statikus tömbként kezelt halmazokkal foglalkoztunk, tehát a 
halmaz elemeit (a tömb tagjait) közvetlenül a felhasználó adja meg, s az elemek az 
eljárás során már nem változnak. A GAMS azonban a dinamikus tömbök 
létrehozását is támogatja (lásd GAMS User's Guide). 

5. ADATOK 

GAMS-ban megfogalmazott szállítási modellünk az adatbevitel mindhárom alapvetı 
formáját - listák, táblázatok, közvetlen értékadás - tartalmazza.  

5.1. ADATBEVITEL LISTÁVAL 

A listával történı adatbevitel a Parameters utasítással történik. Szállítási feladatunk 
Parameters utasítása a következı volt: 
 

Parameters 
a(i) i konzervgyar kapacitasa ladaban 

/  Kmet  350 
Szeged  600 / 

 
b(j) kereslet j piacon ladaban 

/  Sopron 325 
Szhely  300 
Zeg  275 /; 

A fenti utasítás két paramétert (külsı adatot, exogén változót) deklarált, a és b névvel 
illette ıket, deklarálta a hozzájuk tartozó "domain"-eket (tartományokat). (A 
tartomány egy halmaz, amelyre paraméter, változó vagy egyenlet/egyenlıtlenség 
épül.) Az utasítás minden paraméternél magyarázó szöveget is tartalmaz, valamint i 
és j minden elemére megadja a(i) és b(j) értékeit. Teljesen elfogadott, hogy 
utasításainkat több utasításba tördeljük, ahogy az alábbiakban is látható: 
 

Parameter  a(i) i konzervgyar kapacitasa ladaban 
/  Kmet  350 

Szeged  600 /; 
 

Parameter  b(j) kereslet j piacon ladaban 
/  Sopron 325 

Szhely  300 
Zeg  275 /; 

 
Amikor az adatokat listával visszük be, néhány alapvetı szabályt érdemes betartani: 
 
1. A tartományok elemeit és a hozzájuk tartozó paraméterek értékeit lényegében 

tetszés szerint lehet elrendezni. Egy szabályt azonban be kell tartani: a listát 
perjelek (slash) közé kell tenni és a listaelemeket vesszıvel kell elválasztani 
vagy külön sorba kell írni. 

 
2. A lista elemei, a név, a tartomány és a magyarázó szöveg közé nem kell 

pontosvesszıt tenni, hiszen az utasítás minden eleme szükséges a 
deklarációhoz. Ha csak listával akarnánk paramétereket létrehozni, a GAMS 
hibát jelezne. 

 
3. A GAMS fordító rendelkezik egy rendkívül hasznos vonással, az ún. "domain 

checking"-gel (tartományellenırzéssel), amely a tartomány minden eleménél 
megvizsgálja, hogy része-e a hivatkozott halmaznak. Ha például i halmaz 
elemei közé a Zeg városnevet helyesen vettük fel, de egy késıbbi listában 
helytelenül Z.egerszeg néven hivatkozunk rá, a fordító hibát jelez. 

 
4. Minden paraméter alapértelmezett értéke nulla, ezért az értéklistában csak a 

nullától különbözı értékeket kell feltüntetnünk. 
 
5. A skalár tartomány nélküli paraméter. A Scalar utasítás csak egyetlen értéket 

tartalmaz, ahogy azt a szállítási példánkban is láthattuk: 
 

Scalar f kilometerre eso fuvardij ladankent forintban /90/; 
 

Ha a paraméter tartománya két vagy több dimenziós, az értékeket akkor is 
felvehetjük lista formátumban. Ez rendkívül hasznos hiányos (kevés nullától 
különbözı értéket tartalmazó) tömb esetén. 

5.2. ADATBEVITEL TÁBLÁZATTAL 

Az optimalizációs problémáknál gyakran megfigyelhetı, hogy a nagyobb, 
bonyolultabb modellek bemenı adatai viszonylag kis számtáblázatokból 
származnak, így a GAMS nyelv által lehetıvé tett táblázatos adatbevitel megkönnyíti 
a munkánkat. Szállítási problémánk egy kétdimenziós táblázatot tartalmaz: 
 

Table d(i,j) tavolsag kilometerben 
Sopron  Szhely  Zeg 

Kmet   302  279  266 
Szeged   388  354  342 ; 

 
Az utasítás deklarálja d paramétert, meghatározza a hozzátartozó tartományt 
(domain) i és j Descartes-szorzataként. A d-hez tartozó értékeket is az utasításban 
adjuk meg. A táblázatban esetlegesen kihagyott helyeket a GAMS nullaként 
értelmezi. 
 
A listával történı adatbevitelhez hasonlóan itt is mőködik a tartományellenırzés 
(domain checking), tehát a GAMS fordítója megvizsgálja, hogy a táblázat sor-, és 
oszlopnevei tagjai-e a hivatkozott halmaznak/tartománynak. Az összetettebb és a 
kettınél több dimenziót tartalmazó táblázatokról a felhasználói kézikönyvekben 
tájékozódhatunk. 

5.3. ADATBEVITEL KÖZVETLEN ÉRTÉKADÁSSAL 

A közvetlen értékadással történı adatbevitel alapvetıen különbözik a listás és a 
táblázatos adatbeviteltıl, mivel a paraméter deklarálása és az értékadás külön 
utasításokban történik. Szállítási modellünkben a következı közvetlen értékadás 
található: 
 

Parameter c(i,j) ladankenti szallitasi koltseg forintban; 
 

c(i,j) = f*d(i,j) ; 
 
Fontos, hogy elsı sor végén is pontosvesszı található, hiszen e nélkül a GAMS 
fordítója a két sort egy utasításként próbálná értelmezni. (A GAMS természetesen 
képtelen lenne értelmezni a kódot, és hibaüzenetet küldene.) 
 
Az elsı utasítás hatására megtörténik c deklarációja és a hozzá tartozó tartomány 
(domain) kijelölése. Az utasítás magyarázó szöveget is tartalmaz. A második utasítás 
értéket ad c(i,j)-nek. Természetesen ez az értékadás csak akkor lehetséges, ha f és 
d(i,j) értékadása már egy korábbi utasításban megtörtént. 
 
A fenti közvetlen értékadás minden c(i,j) elemre vonatkozik. Ha egy konrét c(i,j) 
elemnek akarunk értéket adni, akkor a tartományneveket idézıjelbe kell tennünk. Pl.: 
 

c('Kmet','Sopron') = 302 ; 
 
Ugyanannak a paraméternek természetesen többször is értéket adhatunk. Minden 
értékadás azonnal felülírja a korábbi értékadást. Ellenben a paramétereket csak 

egyszer deklarálhatjuk, így elkerülhetjük, hogy két eltérı objektumot ugyanolyan 
névvel illessünk. 
 
Az értékadás során a matematikai kifejezések és beépített függvények gazdag 
választékából válogathatunk. A programozáshoz egy keveset már értı 
felhasználóknak nem jelenthet gondot a GAMS használata, azonban a GAMS 
rendelkezik a más programnyelvekben ismeretlen, de a matematikai 
programozásban rendkívül hasznos jellemzıkkel. Például, c(i,j)-nek minden (i,j) 
párra értéket adhatunk ciklusok használata nélkül. 
 
Az alábbi néhány példa a GAMS függvényeinek és standard operátorainak a 
használatát szemlélteti. A használt paramétereket az értékadás elıtt minden esetben 
deklarálni kell! 
 
 csquared = sqr(c); 
 e = m * csquared; 
 w = l / lamda; 
 
 eoq(i) = sqrt ( 2 * demand(i) * ordcost(i) / holdcost(i)); 
 t(i) = min (p(i), q(i) / r(i), log (s(i))); 
 euclidean(i,j) = sqrt (sqr(xi(i) - xi(j) + sqr (x2(i) - x2(j))); 
 present(j) = future(j) * exp(-interest*time(j)); 
 
Az összegzı és a szorzó operátorokról a segédlet késıbbi részében még szó esik.  
 

6. VÁLTOZÓK 

A döntési változókat (endogén változókat) a GAMS-ban megfogalmazott 
modellekben Variables utasítással kell deklarálni. Minden változónak nevet kell 
adni, s ha szükséges, a változót értelmezési tartománnyal (domain) és magyarázattal 
is elláthatjuk. Szállítási modellünkben a Variables utasításra a következı példát 
találjuk: 
 

Variables 
x(i,j) szallitott mennyiseg ladaban 
z forintban mert osszes szallitasi koltseg ; 

 
A fenti utasítás a szállított mennyiséget minden (i,j) párra deklarálta. A z változó 
nem rendelkezik tartománnyal (domain-nel), mivel skalár mennyiség. Minden 
optimalizációs GAMS-modell rendelkezik egy z-hez hasonló változóval, amelynek a 
mennyiségét maximalizálni vagy minimalizálni kell. 

A deklarálás során minden változónak meg kell határoznunk a típusát. Az alábbi 
típusokat választhatjuk: 
 
A változó típusa A változó értelmezési tartománya 
Free 
Positive 
Negative 
Binary 
Integer 

- ∞ ... + ∞  
0 ... + ∞ 
− ∞ ... 0 
0 vagy 1 
0,1,...., 100 
 

 
A változót, amelynek értékét optimalizálni kívánjuk, free típusú skaláris változónak 
kell kiválasztanunk. Szállítási példánkban, z típusát meghagytuk az alapértelmezés 
szerinti free típusnál, de x (i,j) változó értelmezési tartományát nemnegatívvá tettük.  
 

Positive variable x ; 
 
Jegyezzük meg, hogy x típusának megjelölésénél a tartományt (domain) már nem 
kell szerepeltetni, a tartomány minden eleme automatikusan felveszi a változó 
típusát. 
 
A változók alsó és felsı korlátairól, kezdıértékeirıl a késıbbiekben még szó esik. 

7. EGYENLETEK 

A GAMS és a hozzá hasonló matematikai modellezı programnyelvek egyik fı 
erıssége a modellt alkotó egyenletek és egyenlıtlenségek algebrai struktúrájának 
rugalmas, szimultán módon történı létrehozásában rejlik. 

7.1. AZ EGYENLETEK DEKLARÁCIÓJA 

Az egyenleteket külön utasításokkal kell deklarálni, ill. definiálni. A deklaráció 
lényegében azonos a többi GAMS-elemnél megismertekkel. Az Equations utasításnév 
után következik az egyenlet neve, a tartomány (domain) és a magyarázó szöveg. 
Szállítási példánk a következı egyenlet deklaráció(ka)t tartalmazza: 
 

Equations 
 

cost celfuggveny 
supply(i) i gyar kinalati korlatja 
demand(j) j piac kielegitendo kereslete; 

 
A GAMS-nyelvben az Equations utasítással nemcsak egyenleteket, hanem 
egyenlıtlenségeket is létrehozhatunk, tartományra (domainre) való hivatkozással 
vagy anélkül. A továbbiakban az egyenlet szó alatt egyenlıtlenségekre is gondolunk. 

7.2. A GAMS ÖSSZEGZİ ÉS SZORZÓ KIFEJEZÉSEI 

Mielıtt megismerkednénk az egyenletek definíciójával, érdemes néhány szót ejteni a 
GAMS-nyelv összegzı és szorzó kifejezéseirıl. A GAMS-ot úgy tervezték, hogy 
használata a speciális szimbólumokat nem tartalmazó billentyőzetek használata 
mellett is kényelmes legyen (ennek igazából csak a nem latin betős íráshoz tervezett 
billentyőzetnél van jelentısége.) 
 
A GAMS összegzı jelölései egyszerő és összetett kifejezéseknél egyaránt 
használhatók. Az összegzés alapformája egy két argumentumból álló operátor: 
 

Sum(az összegzés indexe, összegzendı) 
 

A két argumentumot minden esetben vesszıvel kell elválasztani, akkor is, ha az elsı 
argumentum zárójelben van. A második argumentum bármilyen, összegzést 
tartalmazó matematikai kifejezés lehet. 
 
Szállítási modellünk az alábbi egyszerő kifejezést tartalmazza: 

 
Sum(j, x(i,j)) 

 
Ez ekvivalens a ∑jxij kifejezéssel. 
 
Egy kicsit összetettebb példa: 
 

Sum((i,j), c(i,j)*x(i,j)) 
 
Hagyományos matematikai jelöléssel: ∑i∑jcijxij. 
 
Utolsó példánk egymásba ágyazott összegzı kifejezéseket szemléltet: 
 
 Sum(i, Sum(j, c(i,j)*x(i,j)))  
 
A GAMS-nyelvben a szorzat pontosan ugyanolyan formátumban írható fel, csak 
Sum helyett Prod operátort kell használnunk. Például a 
 

Prod(j, x(i, j)) 
 

ekvivalens a IIj xij kifejezéssel. 
 
Az összegzı és a szorzó operátorok a paraméterek közvetlen értékadásánál is 
használhatók. 
 

scalar totsupply a gyarak osszkinalata 
totsupply = sum(i, b(i)) ; 

7.3. EGYENLET DEFINÍCIÓ 

Az egyenlet definíció a GAMS legösszetettebb utasítása. Az egyenlet definíció részei 
sorrendben a következık: 
 
• Az egyenlet (egyenlıtlenség) neve 
• A tartomány (domain) 
• A tartományt korlátozó feltétel (nem kötelezı) 
• A ".." jelölés 
• Az egyenlet (egyenlıtlenség) baloldala 
• Reláció operátor: =l=, =e=, =g= 
• Az egyenlet (egyenlıtlenség) jobboldala 
 
A szállítási példa három egyenlet definíciót tartalmaz: 
 

cost.. z =e= sum((i,j), c(i,j)*x(i,j)) ; 
supply(i) .. sum(j, x(i,j)) =l= a(i) ; 
demand(j) .. sum(i, x(i,j)) =g= b(j) ; 

 
Néhány dolgot érdemes megjegyeznünk: 
 
• Az összetett egyenletek az egész tartományokra (domain-ekre) vonatkozó 

definíciók miatt viszonylag könnyen felírhatók. Például a fenti demand korlát a j 
tartomány minden elemére vonatkozik, ahogy ezt a szállítási feladathoz tartozó 
GAMS outputból származó részlet is megmutatja: 

 
demand(Sopron).. 
x(Kmet, Sopron) + x(Szeged, Sopron) =g= 325 ; 
demand(Szhely).. 
x(Kmet, Szhely) + x(Szeged, Szhely) =g= 300 ; 
demand(Zeg).. 
x(Kmet, Zeg) + x(Szeged, Zeg) =g= 275 ; 

 
A GAMS hatékony voltát mutatja, hogy egy ilyen meglehetısen kevés változót 
tartalmazó feladat keresleti korlátját ugyanúgy írjuk fel, mint egy többtízezer 
változós feladatét. Mindkét esetben elegendı, ha a felhasználó az algebrai modellt 
megfogalmazza, az egyes korlátokat a GAMS minden esetben hozzáilleszti a 
modellhez, ahogy azt a fenti output részletben is láthattuk. (Más optimalizációs 
programnyelvekben már a bemeneti oldalon a fenti, kimeneti oldalból kiemelt 
egyenleteket, egyenlıtlenségeket kell felírnunk.) 
 

• A relációs operátorok jelentése a következı: 
 

=l= kisebb vagy egyenlı 
=g= nagyobb vagy egyenlı 
=e= egyenlı 

A "=" és ="=e=" szimbólumokat nem szabad összekeverni. Az "=" jelet csak 
közvetlen értékadásra, a "=e=" szimbólumot csak egyenlet definiálására 
használjuk. 

 
• A változókat az egyenletek bármelyik oldalán szerepeltethetjük. 
 
• Az egyenlet definíció a GAMS bemeneti oldalán bárhol felírható, de az 

egyenletben szereplı változókat, paramétereket, s persze magát az egyenletet az 
egyenlet definiálása elıtt deklarálnunk kell. Az egyenleteket nem kell a 
deklarációnál kialakított sorrendben definiálnunk, bár az áttekinthetıség emellett 
szól. 

8. CÉLFÜGGVÉNY 

A GAMS nem rendelkezik külön célfüggvényt szimbolizáló elemmel. A célfüggvény 
létrehozásához egy free típusú, skaláris változót kell deklarálnunk, a változót egy 
egyenlet definícióba kell beépítenünk, s ezzel lehetıvé válik, hogy a változó betöltse 
a célfüggvény funkcióját. 

9. A MODEL ÉS A SOLVE UTASÍTÁS 

A Model szó alatt egy egyenletgyőjteményt értünk, tehát kijelöljük azt, hogy az 
optimalizációs eljárásnál mely egyenleteket kívánjuk felhasználni. A használandó 
modellt a deklaráció során, a Model utasításnevet követıen névvel látjuk el, majd 
zárójelek (perjelek) között felsoroljuk, hogy a modellünknek mely egyenletek a 
részei. Ha minden korábban definiált egyenletet be kívánunk építeni, a zárójelek 
közé az all szót kell begépelnünk. Szállítási modellünkben egyetlen model utasítás 
szerepel: 
 

Model szallfel /all/; 
 
Az utasítás talán feleslegesnek tőnhet, de rendkívül hasznos azoknak a haladó 
szinten lévı GAMS felhasználóknak, akik egyetlen GAMS programban több 
különbözı modellt alkotnak. Fenti utasításunknak ekvivalens a következıvel: 
 

Model szallfel /cost, supply, demand/; 
 
A tartományok természetesen nem szerepelnek az egyenletek nevei után. 
 
Miután a modellünket megalkotunk, nincs más hátra, mint hogy meghívjuk a 
szükségesnek ítélt solvert (megoldó eljárást). A meghívás a Solve utasítással történik, 
ahogy az a szállítási modellünkben is látható: 
 

solve szallfel using lp minimizing z ; 
 
A solve utasítás a következı formában írható fel: 
 
1. Solve utasításnév 
2. A megoldandó modell neve 

3. A using kulcsszó 
4. Az alkalmazandó megoldó (solver) eljárás megnevezése 

lp - lineáris programozás 
nlp - nemlineáris programozás 
mip - kevert egészértékő programozás 
minlp - kevert egészértékő nemlineáris programozás 

 stb. (számos egyéb megoldó eljárás alkalmazható) 
5. A minimizing vagy maximizing kulcsszó 
6. Az optimalizálni kívánt változó 

10. A DISPLAY UTASÍTÁS 

A .gms kiterjesztéső input fájl elkészítése után az állományt lefuttatjuk. A GAMS 
fordító létrehoz egy .lst kiterjesztéső output fájlt, amely tartalmazza a modellünk 
megoldását, ill. minden, a modellel kapcsolatos információt. A modell szintaktikai 
hibái, a hibákhoz tartozó számkódokkal egy külön fájlban jelennek meg. Az output 
fájl meglehetısen összetett adathalmazából a display utasítás segítségével érdemes 
néhány adatot kiemelni. Esetünkben a 
 
 Display x.l, x.m; 
 
utasítás x(i,j) primál- (.l) és duálváltozóit (.m) jeleníti meg könnyen áttekinthetı 
formában. 
 

 
   
2.1 ábra  A szállítási modellhez kapcsolódó futási információk GAMSIDE 20.7 Student felületen 

11. AZ ".LO, .L, .UP, .M" ADATBÁZIS 

A GAMS tartalmaz egy kis adatbázist, ahol a GAMS minden egyes változóhoz és 
egyenlethez egy rekordot rendel hozzá. Minden rekord négy mezıt tartalmaz: 
 
 .lo = alsó határ 
 .l = primálváltozó 
 .up = felsı határ 
 .m = duálváltozó 
 
A fenti mezıkhöz a felhasználónak teljes olvasási/írási jogosultsága van. Ez 
elsısorban a professzionális alkalmazásoknál elınyös. 
 

12. A VÁLTOZÓK HATÁRAINAK ÉS KEZDİÉRTÉKEINEK MEGADÁSA 

A változók alsó és felsı korlátja a változó típusából (free, positive, negative, binary, 
integer) automatikusan következik, de ezeket a korlátokat a GAMS felhasználó 
bármikor felülírhatja. Néhány lehetséges kódrészlet: 
 
 x.up(i,j) = capacity(i,j) ; 

x.lo(i,j) = 10.0 ; 
x.up('Kmet','Sopron') = 1.2*capacity('Kmet','Sopron') ; 

 
Természetesen azt feltételezzük, hogy az elsı és a harmadik példában lévı 
capacity(i,j) paramétert már korábban deklaráltuk és értékkel láttuk el. Ezeket az 
utasításokat a változó deklaráció után, de a Solve utasítás elıtt kell elhelyeznünk. A 
jobboldalon minden matematikai kifejezés felhasználható. 
 
Nemlineáris programozási feladatoknál nagyon fontos, hogy a modellezı a lehetı 
legszőkebb értelmezési tartományt állítsa be az alsó és/vagy a felsı korlát 
segítségével, és megkönnyítse ezzel a megoldó eljárás (solver) munkáját. Nagyon 
hasznos, ha megadunk egy, a várható megoldástól nem túlságosan messze esı 
kezdıértéket az .l és vagy az .m mezık beállításával, amely az optimum keresésénél 
kiindulási alapul szolgálhat.  Egy lehetséges kódrészlet: 
 

quantity.l(i) = 0.5*eoq(i) ; 
 
(Az alapértelmezett kezdıérték általában nulla. Ha a nulla nem része az alsó és felsı 
korláttal beállított értelmezési tartományunknak, akkor az alapértelmezett 
kezdıérték a nullához legközelebb esı érték.) 
 
Az .lo és .up mezık teljes egészében a GAMS felhasználók felügyelete alatt állnak. 
Ezzel szemben az .l és az .m mezık kezdeti értékét ugyan a felhasználó állítja be, 
azonban a mezıket az aktuális megoldó eljárás (solver) szabályozza, végsı értékük a 
modell és a megoldó eljárás függvénye. (Felhívnánk a figyelmet arra, hogy a korábbi 
display utasításnál az .l és az .m mezık a primál és duál változók megjelenítését 
szolgálta!) 

13. A GAMS OUTPUT 

A GAMS output (kimeneti fájl) része nagyon sok elembıl áll, és meglehetısen 
informatív. Az egyes összetevık részletes tárgyalását a terjedelmi korlátok miatt 
elhagyjuk, a különbözı felhasználói kézikönyvek részletesen foglalkoznak a 
kimeneti fájl szerkezetével. 

14.  OPERÁCIÓKUTATÁSI PÉLDÁK 

A GAMS nyelvet elsısorban lineáris programozási feladatok megoldására hozták 
létre, azonban mostanra a GAMS nyelv segítségével minden operációkutatási 
feladatot megoldhatunk, az alábbiakban ezek közül mutatok be néhányat. 

14. 1. MINIMUMFELADAT MEGOLDÁSA 

Az operációkutatásból jól ismert minimumfeladat megoldására az alábbi GAMS 
program nyújt példát.  A minimumfeladatnál, az elızı szállítási feladathoz 
hasonlóan lp (lineáris programozás) solvert (megoldó eljárást) alkalmazunk. Talán 
nem felesleges felírni a lineáris programozási feladatok matematikai alakját. 
 
Min. vagy max.  cx 
         Feltételek: Ax ≥ b 
   L   ≤ x ≤  U 
 
ahol x változóvektor valós szám, cx a célfüggvény, Ax ≥ b a feltételeket, L és U a 
változók alsó és felsı határait jelöli. (L általában 0) A konkrét feladat matematikai 
modelljét - egyszerősége miatt – a GAMS bemeneti fájl megjegyzései közé szúrtam 
be. Minimumfeladatunk megoldása tehát a következı: 
 

*MINIMALIZALNI KELL AZ 
*F(X)= 2*X1 - X2 
*KORLATOZO FELTETELEK: 
*X1-2X2-2=0 
*X1+X2-2>=0 
*2X1 + 3X2 -12<=0 
*X1>0 
*X2>0 

 
VARIABLES X1, X2, F; 

 
POSITIVE VARIABLES X1, X2; 

 
EQUATIONS CELFGV, EGYENL1, NAGYOBB1, KISEBB1; 

 
CELFGV.. F =E= 2*X1 - X2; 
EGYENL1.. X1-2*X2 =E= 2; 
NAGYOBB1.. X1+X2 =G= 2; 
KISEBB1.. 2*X1 + 3*X2 =L= 12; 

 
MODEL OPTIM /ALL/; 

 
SOLVE OPTIM USING LP MINIMIZING F; 

 
DISPLAY F.L; 

DISPLAY X1.L; 
DISPLAY X2.L; 
DISPLAY CELFGV.L; 
DISPLAY EGYENL1.L; 
DISPLAY KISEBB1.L;  

 
Természetesen a maximumfeladatok, normálfeladatok, módosított normálfeladatok, 
általános feladatok megoldásánál a fenti kódrészlet kis módosítással felhasználható. 

14.2. KÖLTSÉGMINIMALIZÁSI FELADAT MEGOLDÁSA 

Következı példánk a mikrogazdaságtanból jól ismert Copp-Douglas termelési 
függvényt feltételezı termelési költségminimalizálási nlp (nemlineáris) 
programozási példa. 
 
A nemlineáris problémák matematikai alakja: 

       Min.  f(x) 
         Feltételek: g(x) ≤ 0 
   L   ≤ x ≤  U 
 
ahol x változóvektor valós szám, f(x) a célfüggvény, g(x) a korlátokat, L és U a 
változók alsó és felsı határait jelöli. 
 
A feladat megoldásánál egyedül a hatványra emelésnél használt '**' jelölés jelenthet 
újdonságot. A konkrét feladat matematikai modelljét - egyszerősége és a 
mikrogazdaságtanból való ismertsége miatt – a GAMS bemeneti fájl megjegyzései 
közé szúrtam be: 
 

*K,L,C TERMELESI KTSG MINIMALIZALAS 
*MIN C= 4*K + 5*L 
*KORLAT 10* K(HATVANY 0.5) * L(HATVANY 0.5) >= 1000 
* K>=10 L>=10 

 
VARIABLES K, L, C; 

 
POSITIVE VARIABLES K,L; 
K.L=10; 
L.L=10; 

 
EQUATIONS 
OBJ KOLTSEGFUGGVENY 
P TERMELES MENNYISEGE; 

 
OBJ.. C =E= 4*K + 5*L; 
P.. 10*(K**0.5 * L**0.5) =G= 1000; 

 
MODEL MINOBJ /ALL/; 

 
SOLVE MINOBJ USING NLP MINIMIZING C; 

14.3. HASZNOSSÁGI FELADAT MEGOLDÁSA 

Kéttermékes hasznossági modellünk szintén a mikrogazdaságtanból átemelt 
nemlineáris programozási példa. A Solver munkájának könnyítése érdekében X és Y 
változókhoz alsó korlátokat (határokat) rendeltünk. Ez nemlineáris programozási 
feladatoknál gyakori. A hasznossági modell a standard mikroökonómiából ismert, 
így matematikai modelljét elhagytam. 
 

VARIABLES 
X 
Y 
U; 
X.L=1; 
Y.L=1; 

 
EQUATIONS 

 
OBJ     HASZNOSSAG FUGGVENY 
RP       KOLTSEGVETESI KORLAT; 

 
OBJ.. U =E= (X+2)*(Y+1); 
RP.. 4*X + 6*Y =E= 130; 

 
MODEL HASZNMAX /OBJ, RP/; 

 
SOLVE HASZNMAX USING  NLP MAXIMIZING U; 

 

14. 4. HOZZÁRENDELÉSI FELADAT MEGOLDÁSA  

Hozzárendelési feladatunk a szállítási feladattal rokon lineáris programozási feladat. 
Általában hozzárendelési probléma címen foglalják össze azokat a szállítási 
feladatokat, amelyekben a feladóhelyek mindegyikén egyetlen termék elszállításáról 
van szó és a felvevıhelyek igénye is egyetlen termék. 
 
Klasszikus példa erre a következı feladat (a hozzárendelési feladat legegyszerőbb 
formája): n számú feladat elvégzésére n munkás áll rendelkezésre úgy, hogy egy 
munkás csak egy feladatot végez el, és egy feladattal csak egy munkás foglalkozik. 
(A munkások és a feladatok kölcsönösen és egyértelmően egymáshoz lesznek 
rendelve.) Matematikai jelölésekkel: 
 

cij = i-edik munkás j-edik feladatot mekkora cij költséggel végzi 
xij = i-edik munkás j-edik feladatot elvégzi v. sem 

A GAMS bemeneti fájlhoz tartozó kód: 

 
SETS 

           GEP /GEP1, GEP2, GEP3, GEP4/ 
           MUNKAS /MUN1, MUN2, MUN3, MUN4/; 
 

TABLE T(I,J) IDO 
           UN1    MUN2   MUN3   MUN4 

GEP1      14      5        8        7 
GEP2      2        12       6        5 
GEP3      7        8        3        9 
GEP4      2        4        6        10; 

 
VARIABLES 
X(I,J) 
Z; 
POSITIVE VARIABLES X(I,J); 

 
EQUATIONS 

 
OBJ 
KORL1 (I) 
KORL2 (J); 

 
OBJ.. Z =E= SUM((I,J), T(I,J)*X(I,J)); 
KORL1(I).. SUM(J, X(I,J)) =E= 1; 
KORL2(J).. SUM(I, X(I,J)) =E= 1; 

 
MODEL HOZZAR/ALL/; 

 
SOLVE HOZZAR USING LP MINIMIZING Z; 

 
DISPLAY X.L, Z.L; 

14. 5. EGÉSZÉRTÉKŐ PROGRAMOZÁSI FELADAT INPUT FÁJLJA 

Az egészértékő programozási feladat (MIP solver) matematikai modellje a 
következı: 
 
A nemlineáris problémák matematikai alakja: 

       Min.  f(x) 
         Feltételek: g(x) ≤ 0 
   L   ≤ x ≤  U 
 
ahol x változóvektor valós szám, f(x) a célfüggvény, g(x) a korlátokat, L és U a 
változók alsó és felsı határait jelöli. A konkrét feladat matematikai modelljét - 
egyszerősége miatt – a GAMS bemeneti fájl megjegyzései közé szúrtam be: 

*EGESZERTEKU PROGRAMOZAS 
*MAXIMALIZALNI Z= 4X1+X2 
*KORLATOK: 
*2X1+X2=<5 
*2X1+3X2=5 
*X1>=0 
*X2>=0 
*X1,X2 EGESZ 

 
VARIABLES X1, X2, Z; 

 
INTEGER VARIABLES X1, X2; 

 
EQUATIONS OBJ, EGY1, EGY2; 

 
OBJ.. Z =E= 4*X1 + X2; 
EGY1.. 2*X1 + X2 =L= 5; 
EGY2.. 2*X1 + 3*X2 =E= 5; 

 
MODEL EGESZLP /ALL/; 

 
SOLVE EGESZLP USING MIP MAXIMIZING Z; 

14. 6. EGY KLASSZIKUS MUNKAERİGAZDÁLKODÁSI MODELL  

Az alábbi, munkaerı-gazdálkodási modell a klasszikus, Dantzig által publikált 
modell egyszerősített változata. A matematikai modell az Equations rész egyenletei-
egyenlıtlenségei alapján egyértelmő, külön magyarázatot nem igényel. 
 

Set       i      idoszak (hetben) / 1*5 / 
 

Scalars    term   termelekenyseg      (aruegysegben munkasonkent)  /     8  / 
            kap    oktato kapacitasa   (egy oktatora eso dolgozo)    /     6  / 

             ber    ber                   (euroban dolgozonkent)        /   100  / 
 

Parameter  ka(i)  kezdo arukeszlet (aruegysegben)       / 1  10  / 
             kl(i)  alkalmazottak indulo letszama         / 1  20  / 
             vs(i)  vegkielegites    ( euroban )           / 5 100  / 

            ke(i)  kereslet         (aruegysegben)       / 1 100, 2 200, 3 300 
                                                            4 400, 5 200 / 

 
Positive Variables 

   t(i)     termelesi szint i idoszakban           (aruegysegben) 
a(i)    arukeszlet i idoszakban (raktar)    (aruegysegben) 

   u(i)    kielegitetlen kereslet i idoszakban (aruegysegben) 
   p(i)    potencialis dolgozok szama 
   h(i)    alkalmazasra kerulok szama 
   f(i)     elbocsatottak szama 
 

Variables 
   c        osszkoltseg                           (euroban) 
 

Equations 
ae(i)    aruppiaci egyensuly 
me(i)  munkapiaci egyensuly idoszaki osszehasonlitasban 
met(i) munkapiaci egyensuly a termeles figyelembevetelevel 

   cel       celfuggveny                           (euroban); 
 

ae(i).. a(i) =e= a(i-1) + t(i) - ke(i) - u(i-1) + u(i) + ka(i); 
 

me(i).. p(i) =e= p(i-1) - f(i) + h(i) + kl(i); 
 

met(i).. p(i) =g= t(i)/term + (1 + 1/kap)*h(i); 
 

cel.. c =e= sum(i, 10*a(i) + 30*u(i) + (ber + vs(i))*p(i) ); 
 

Model munkaero munkaero kepzesi modell / all /; 
 

Solve munkaero minimizing c using lp; 

 

14. 7. EGY OPTIMÁLIS NÖVEKEDÉSI MODELL MEGOLDÁSA 

Az alábbi, Chakravartytól származó – a Ramsey-modellhez hasonló - növekedési 
modell bemeneti fájlja a GAMS nyelv újabb eszközeivel, lehetıségeivel ismerteti meg 
az olvasót. Az új elemek bemutatása elıtt az eddigieknél bonyolultabb közgazdasági 
modellt érdemes matematikai jelölésekkel is felírni. A megoldás során néhány 
esetben kénytelen voltam a jelölésektıl eltérni. Az általam közölt kód a GAMS 
fordító szempontjából, a matematikai modell közgazdasági-matematikai 
szempontból tekinthetı teljesnek. (Valójában a két modell ekvivalens egymással.) Az 
egyes jelöléseken túl a GAMS kód elemei is némileg eltérı jelentéssel bírnak. A 
modell algebrai jelölésekkel a következı: 
 

Yt = kibocsátás (jövedelem, output)  t idıszakban 
α  = termelési függvény paraméter 
K0 = tıkeállomány 0 idıszakban 
Kt =  tıkeállomány t idıszakban 
β = tıkekitevı - tıkehatvány 
Ct =  fogyasztás t idıszakban 
U(Ct) hasznosság t idıszakban a fogyasztás függvényében 
τ = hasznossági függvény paramétere – rugalmasság 
J = teljesítmény – kibocsátás 
dis = diszkontfaktor 
ρ = (jóléti) diszkonttényezı 

 
 

Felt. Kt+1 = Kt + Yt – Ct 
 
  Kt+1 = Kt + αKtβ – Ct 
 
  Kn ≥ K* 

Max. J 
 
A GAMS bemeneti fájlban a bevezetı $Title utasítása utáni, címként funkcionáló 
részt és az $Ontext - $Offtext között elhelyezkedı magyarázó részt a fordító figyelmen 
kívül hagyja. Az $Ontext - $Offtext között olyan, a megértést szolgáló szöveges részeket 
helyezhetünk el, amelyeknél a magyar nyelvnek megfelelı közép-európai kódolású 
betőkészletet, így ékezetes betőket is használhatunk! 
 
A Sets-nél szereplı bázisidıszak és tárgyidıszak dinamikus tömbök definíciója több, 
eddig nem tárgyalt elemet tartalmaz: 
 
• Az eq numerikus reláció operátor két számértéket hasonlít össze, egyezés esetén a 

visszatérési érték True, ellenkezı esetben False. (Igaz-Hamis, 1-0)  
 
• Alapértelmezés a szerint a Set segítségével definiált halmazok elemei számértékkel 

– még ha az elemek csupa számjegybıl is állnak – nem rendelkeznek. Az ord 
operátor az egydimenziós, rendezett elemekbıl álló, statikus halmaz (tömb) 
elemeit számértékké alakítja át, mégpedig úgy, hogy visszaadja az elem relatív 
pozícióját. A példában t=0 esetén ord(t)=1, t=1 esetén ord(t)=2 stb. Más 
programnyelvektıl eltérıen a halmazelemek-tömbelemek sorszámozása nem 0-
val, hanem 1-gyel kezdıdik. A card operátor bármely Set utasítással létrehozott 
halmaz elemszámát adja vissza. 

 
• A Yes kulcsszót (a No kulcsszóval együtt) a más programnyelvekból ismert 

dinamikus tömbök definiálására használhatjuk fel a  
 
set_név (domain_név | domain_elem) = yes | no ; 
 

szintaxis alapján. A Yes után szereplı $ jel segítségével a Yes kulcsszót logikai 
utasítássá alakítjuk, esetünkben pl. az elsı dinamikus tömbnél - itt most „halmaz” 
helyett a többi programozási nyelvvel történı összehasonlíthatóság érdekében  a 
„tömb” kifejezést használom – ord(t) akkor lesz tb(t) dinamikus tömb tagja, ha 
ord(t)=1. 
 

• A fentiek alapján tb(t)=0, tt(t)=20. 
 
Növekedési modellünk GAMS bemeneti fájlja tehát a következı: 
 

$Title Optimalis novekedesi modell 
 

$Ontext 
 

Egyszektoros nemlineáris növekedési modell 
 

Kendrick, D, and Taylor, L, Numerical methods and Nonlinear Optimizing 

models for Economic Planning. In Chenery, H B, Ed, Studies of 
Development Planning. Harvard University Press, 1971. 

 
Chakravarty, S, Optimum Savings with a Finite Planning Horizon. 
International Economic Review 3 (1962), 338-355. 

 
$Offtext 

 
 Sets   t      idotav     / 0*20 / 

         tb(t)  bazisidoszak 
        tt(t)  targyidoszak; 

 
 tb(t) = yes$(ord(t)eq 1); tt(t) = yes$(ord(t) eq card(t)); 
Display tb,tt; 

 
 Scalars   delt  ertekcsokkenesi rata        /  .05  / 

             beta tokehatvany                 /  .75  / 
a       hatekonysag 

            r       munkaeronovekedesi rata   /  .025 / 
            tau    rugalmassag                 /  .9   / 

             z       technikai haladas           /  .01  / 
           rho   joleti diszkonttenyezo      /  .03  / 

          y0     kezdo jovedelem            / 4.275 / 
            k0     kezdotoke                   / 15.0  / 

 
Parameters  dis(t)  diszkontfaktor 

               alfa(t)   termelesi fuggveny ; 
 

a = y0/k0**beta; 
dis(t)   = (1+rho)**(1-ord(t))/(1-tau); 
alfa(t) = a*(1+r*(1-beta)+z)**(ord(t)-1); 
Display a, dis, alfa ; 

 
Variables   c(t)   fogyasztas 

               y(t) jovedelem kibocsatas 
               k(t)   tokeallomany 

              j     makrookonomiai kibocsatas novekedes 
 

Equations   kb(t)  tokemerleg 
               yd(t)  jovedelem definicio 
               jd     makrookonomiai kibocsatas definicio ; 
 

jd..  j =e= sum(t, dis(t-1)*c(t-1)**(1-tau)); 
 

yd(t)..  y(t) =e= alfa(t)*k(t)**beta; 
 

kb(t+1)..  k(t+1) =e= y(t) - c(t) + (1-delt)*k(t); 
 

y.l(t) = y0*(1.06)**(ord(t)-1);  k.l(t) = (y.l(t)/alfa(t))**(1/beta); 
 
c.l(t) = y.l(t) + (1-delt)*k.l(t) - k.l(t+1);   display c.l, k.l, y.l; 

 
k.lo(t) = 1; y.lo(t) = 1; c.lo(t) =1; y.fx(tb) = y.l(tb); y.fx(tt) = y.l(tt); 

 
Model novekedes / all /; 
Solve novekedes maximizing j using nlp; 

Parameter report solution summary ; 
report(t,"k")  = k.l(t); report(t,"y") = y.l(t); report(t,"c") = c.l(t); 
report(t,"s-rate") = (y.l(t)-c.l(t))/y.l(t); 

  Display report; 
 

14. 8. PORTFOLIÓ-ANALÍZIS  

A portfolió analízis a GAMS-modellek egyik kedvelt alkalmazási területe, az alábbi 
modell erre nyújt egy rendkívül leegyszerősített példát: 
 

i = Részvények (A részvény, B, részvény, C részvény) 
 

B = Befektetésre szánt összeg 
R = Elvárt hozam 
ri  = i részvény várható hozama 
σij = i és j részvény hozama közötti kovariancia 

 
xi = i részvénybe befektetésre kerülı összeg (e Ft) 

 

∑∑
= =

σ
3

1i

3

1j
ijji xxmin  

 

,Bx.felt
3

1i
i∑

=

≤  

 

∑
=

≥
3

1i
ii ,Rxr  

 

0x i ≥  

 
B = 1000.00, R = 50.00 

 
σij 1. részvény      2. részvény      3. részvény 

1. részvény      
2. részvény      
3. részvény 

0.017087987     0.003298885     0.001224849 
0.003298885     0.005900944     0.004488271 
0.001224849     0.004488271     0.063000818 

ri 0.026                 0.008                 0.074 

 
A feladat GAMS kódja: 
 

sets i/1*3/ 
alias (i,j); 

 
parameter rbar(i) 
/ 
1        0.0260023 
2        0.008100891 
3        0.073774971 

/; 
 

table sigma(i,j) 
         1                        2                        3 
1        0.017087987     0.003298885     0.001224849 
2        0.003298885     0.005900944     0.004488271 
3        0.001224849     0.004488271     0.063000818 
; 

 
scalar  B /1000.00/ 

         R /50.00/ 
 

positive variables x(i); 
free variable z; 

 
equations obj, c1, b1; 

 
obj.. z =e= sum(i, sum(j, sigma(i,j)*x(i)*x(j))); 
b1.. sum(i, x(i)) =l= B; 
c1.. sum(i, rbar(i)*x(i)) =g= R; 

 
model portfolio /all/; 

 
solve portfolio minimizing z using nlp; 

 
display x.l; 

MELLÉKLET  
A SZÁLLÍTÁSI FELADAT OUTPUT FÁJLJA 

